Computergestuetzte Mathematik zur Analysis

Lektion 13 (28. Januar)

```
> restart:
```

Gewöhnliche Differentialgleichungen

```
> Dgl := diff(y(x), x) = y(x);
```

$$Dgl := \frac{d}{dx}\, y(x) = y(x) \tag{1.1}$$

```
> dsolve(Dgl, y(x));
```

$$y(x) = \_C1\, e^{x} \tag{1.2}$$

```
> # falls Maple eine Loesung findet, so werden alle Loesungen ausgegeben
> # Integrationskonstante _C1
> AB := y(0) = A;
```

$$AB := y(0) = A \tag{1.3}$$

```
> dsolve({Dgl, AB}, y(x));
```

$$y(x) = A\, e^{x} \tag{1.4}$$

```
> Dgl := diff(y(x),x,x) + 2*m* diff(y(x),x) + o^2*y(x);
```

$$Dgl := \frac{d^2}{dx^2}\, y(x) + 2\, m \left(\frac{d}{dx}\, y(x)\right) + o^2\, y(x) \tag{1.5}$$

```
> sol := dsolve(Dgl,y(x));
```

$$sol := y(x) = \_C1\, e^{\left(-m+\sqrt{m^2-o^2}\right)x} + \_C2\, e^{\left(-m-\sqrt{m^2-o^2}\right)x} \tag{1.6}$$

```
> sol := rhs(dsolve(Dgl,y(x)));
```

$$sol := \_C1\, e^{\left(-m+\sqrt{m^2-o^2}\right)x} + \_C2\, e^{\left(-m-\sqrt{m^2-o^2}\right)x} \tag{1.7}$$

```
> Dsys := { diff(w(x),x) + 2*m*w(x) + o^2*y(x)=0, diff(y(x),x) - w(x) =0 } ;
```

$$Dsys := \left\{\frac{d}{dx}\, y(x) - w(x) = 0,\ \frac{d}{dx}\, w(x) + 2\, m\, w(x) + o^2\, y(x) = 0\right\} \tag{1.8}$$

```
> syssol :=dsolve(Dsys);
```

$$syssol := \left\{w(x) = \_C1\left(-m+\sqrt{m^2-o^2}\right) e^{\left(-m+\sqrt{m^2-o^2}\right)x} + \_C2\left(-m-\sqrt{m^2-o^2}\right) e^{\left(-m-\sqrt{m^2-o^2}\right)x},\ y(x) = \_C1\, e^{\left(-m+\sqrt{m^2-o^2}\right)x} + \_C2\, e^{-\left(m+\sqrt{m^2-o^2}\right)x}\right\} \tag{1.9}$$

```
> sol:=rhs(syssol[2]);
```

$$sol := \_C1\, e^{\left(-m+\sqrt{m^2-o^2}\right)x} + \_C2\, e^{-\left(m+\sqrt{m^2-o^2}\right)x} \tag{1.10}$$

```
> Dgl := diff(y(x),x) - y(x) + x^3 - 3*x + 2 =0;
```

$$Dgl := \frac{d}{dx} y(x) - y(x) + x^3 - 3x + 2 = 0 \quad \textbf{(1.11)}$$

```
> Lsg := dsolve(Dgl, y(x));
```

$$Lsg := y(x) = x^3 + 3x^2 + 3x + 5 + \_C1\, e^x \quad \textbf{(1.12)}$$

```
> eval(Dgl, Lsg);
```

$$0 = 0 \quad \textbf{(1.13)}$$

```
> # Ueberpruefen der Loesung durch Einsetzen in DGL
> eval(Lsg, x = 0);
```

$$y(0) = 5 + \_C1 \quad \textbf{(1.14)}$$

```
> K1 := solve(eval(%, y(0)=1), _C1);
```

$$K1 := -4 \quad \textbf{(1.15)}$$

```
> # Anfangsbedingung gegeben, Konstante ausrechnen
> eval(y(x), {Lsg, _C1=K1});
```

$$x^3 + 3x^2 + 3x + 5 + \_C1\, e^x \quad \textbf{(1.16)}$$

```
> # klappt nicht !
```

```
> f1 := subs(Lsg, _C1=K1, y(x));
```

$$f1 := x^3 + 3x^2 + 3x + 5 - 4\, e^x \quad \textbf{(1.17)}$$

```
> Lsg2 := dsolve({Dgl, y(0) = 1/2}, y(x));
```

$$Lsg2 := y(x) = x^3 + 3x^2 + 3x + 5 - \frac{9}{2}\, e^x \quad \textbf{(1.18)}$$

```
> f2 := rhs(Lsg2);
```

$$f2 := x^3 + 3x^2 + 3x + 5 - \frac{9}{2}\, e^x \quad \textbf{(1.19)}$$

```
> Lsg3 := dsolve({Dgl, y(0) = 3/2}, y(x));
```

$$Lsg3 := y(x) = x^3 + 3x^2 + 3x + 5 - \frac{7}{2}\, e^x \quad \textbf{(1.20)}$$

```
> f3 := rhs(Lsg3);
```

$$f3 := x^3 + 3x^2 + 3x + 5 - \frac{7}{2}\, e^x \quad \textbf{(1.21)}$$

```
> # Loesung der verschiedenen AWA zeichnen
> plot([f1, f2, f3], x = -1.5 .. 2, thickness = 2);
```




```
> pl1 := %:
> with(plots):
> # Richtungsfeld der DGL, nach diff(y(x),x) aufloesen
> isolate(Dgl, diff(y(x), x));
```

$$\frac{\mathrm{d}}{\mathrm{d}x}\, y(x) = y(x) - x^3 + 3\,x - 2 \tag{1.22}$$

```
> v := [1, rhs(%)];
```

$$v := \left[1,\, y(x) - x^3 + 3\,x - 2\right] \tag{1.23}$$

```
> # Vektorfeld ist tangential an den Graphen jeder Loesung
> pl2 := fieldplot(v, x = -1.5 .. 2, y = -2 .. 5, thickness  = 2)
  :
> display({pl1, pl2});
```




```
> # Loesungen verlaufen "entlang" des Vektorfeldes
```

getrennte Variable

```
> restart:
> Dgl := diff(y(x), x) = exp(y(x)) * sin(x);
```

$$Dgl := \frac{d}{dx}\, y(x) = e^{y(x)} \sin(x) \tag{2.1}$$

```
> # Loesung nur lokal auf einem Intervall definiert
> Lsg := dsolve(Dgl);
```

$$Lsg := y(x) = -\ln(\cos(x) - \_C1) \tag{2.2}$$

```
> eval(Dgl, Lsg);
```

$$\frac{\sin(x)}{\cos(x) - \_C1} = \frac{\sin(x)}{\cos(x) - \_C1} \tag{2.3}$$

```
> f := rhs(Lsg);
```

$$f := -\ln(\cos(x) - \_C1) \tag{2.4}$$

```
> eval(f, x=0);
```

$$-\ln(1 - \_C1) \tag{2.5}$$

```
> K1 := solve(% = -1/2, _C1);
```

$$K1 := -e^{\frac{1}{2}} + 1 \quad (2.6)$$

```
> f1 := rhs(eval(Lsg, _C1=K1));
```

$$f1 := -\ln\left(\cos(x) + e^{\frac{1}{2}} - 1\right) \quad (2.7)$$

```
> plot(f1, x = -2 .. 6, thickness = 2);
```




```
> # Logarithmus nicht ueberall definiert
> # Argument des Logarithmus untersuchen
> tmp := cos(x)+exp(1/2)-1;
```

$$tmp := \cos(x) + e^{\frac{1}{2}} - 1 \quad (2.8)$$

```
> a := solve(tmp = 0, x);
```

$$a := \pi - \arccos\left(e^{\frac{1}{2}} - 1\right) \quad (2.9)$$

```
> evalf(a);
```

$$2.276699288 \quad (2.10)$$

```
> simplify(subs(x=-a,tmp));
```

$$0 \quad (2.11)$$

```
> # Funktion f1 ist nur auf dem Intervall (-a,a) eine Loesung der AWA
```

mehrere Lösungen einer AWA

```
> restart:
> # Satz von Picard-Lindeloef : AWA unter gewissen Voraussetzungen lokal eindeutig loesbar
> g := (x^2 + 2*x)*y(x)*exp(y(x)^2)*diff(y(x), x) - 1;
```

$$g := (x^2 + 2\,x)\; y(x)\; e^{y(x)^2} \left(\frac{d}{dx}\, y(x)\right) - 1 \tag{3.1}$$

```
> Lsgn := dsolve({g = 0, y(2) = 0}, y(x));
```

$$Lsgn := y(x) = \sqrt{\ln\left(1 - \ln\left(\frac{1}{2}\,\frac{x+2}{x}\right)\right)},\; y(x) = -\sqrt{\ln\left(1 - \ln\left(\frac{1}{2}\,\frac{x+2}{x}\right)\right)} \tag{3.2}$$

```
> # fuer die AWA y(2)=0 gibt es zwei verschiedene Loesungen
> g1 := unapply(rhs(Lsgn[1]), x): g2 := unapply(rhs(Lsgn[2]), x):
> simplify(eval(g,Lsgn[1])), simplify(eval(g,Lsgn[2]));
```

$$0,\, 0 \tag{3.3}$$

```
> g1(2), g2(2);
```

$$0,\, 0 \tag{3.4}$$

```
> Lsg := dsolve({g = 0, y(2) = 2/5}, y(x));
```

$$Lsg := y(x) = \sqrt{\ln\left(e^{\frac{4}{25}} - \ln\left(\frac{1}{2}\,\frac{x+2}{x}\right)\right)} \tag{3.5}$$

```
> # die AWA y(2)=2/5 hat eindeutige Loesung
> f1 := rhs(Lsg);
```

$$f1 := \sqrt{\ln\left(e^{\frac{4}{25}} - \ln\left(\frac{1}{2}\,\frac{x+2}{x}\right)\right)} \tag{3.6}$$

```
> simplify(eval(g,Lsg));
```

$$0 \tag{3.7}$$

```
> simplify(eval(f1,x=2));
```

$$\frac{2}{5} \tag{3.8}$$

```
> # Definitionsbereich von f1 bestimmen
> a := solve(f1 = 0, x);
```

$$a := \frac{2}{2\, e^{e^{\frac{4}{25}} - 1} - 1} \tag{3.9}$$

```
> with(plots):
> # Richtungsfeld der DGL
> v := <1, 1/((x^2 + x)*y*exp(y^2))>;
```

(3.10)

$$v := \left[\begin{array}{c} 1 \\ \frac{1}{(x^2+x)\, y\, e^{y^2}} \end{array}\right] \quad \textbf{(3.10)}$$

```
> # Vektorfeld normieren
> N := sqrt(v[1]^2+v[2]^2);
```

$$N := \sqrt{1 + \frac{1}{(x^2+x)^2\, y^2\, (e^{y^2})^2}} \quad \textbf{(3.11)}$$

```
> w := v/N;
```

$$w := \left[\begin{array}{c} \frac{1}{\sqrt{1 + \frac{1}{(x^2+x)^2\, y^2\, (e^{y^2})^2}}} \\ \frac{1}{\sqrt{1 + \frac{1}{(x^2+x)^2\, y^2\, (e^{y^2})^2}}\, (x^2+x)\, y\, e^{y^2}} \end{array}\right] \quad \textbf{(3.12)}$$

```
> p := fieldplot(w, x = 1.45..3, y = -0.6..0.6):
> qg := plot([g1, g2], 2..3,colour=[red,green]):
> qf := plot(f1, x=a..3,colour=yellow):
> display({p, qf, qg}, axes = frame);
```




```
> # Beispiel mit unendlich vielen Lösungen
> restart:
> Dgl := diff(y(x),x)*sin(x) = 2*y(x)*cos(x);
```

$$Dgl := \left(\frac{d}{dx}\, y(x)\right) \sin(x) = 2\, y(x) \cos(x) \quad (3.13)$$

```
> AB := y(0) = 0;
```

$$AB := y(0) = 0 \quad (3.14)$$

```
> # keine eindeutige Loesung
> Lsg := dsolve({Dgl, AB}, y(x));
```

$$Lsg := y(x) = -\frac{1}{2}\, \_C1 \cos(2\, x) + \frac{1}{2}\, \_C1 \quad (3.15)$$

```
> simplify(eval(Dgl,Lsg));
```

$$2\, \_C1 \cos(x) \sin(x)^2 = 2\, \_C1 \cos(x) \sin(x)^2 \quad (3.16)$$

```
> f := rhs(Lsg);
```

$$f := -\frac{1}{2}\, \_C1 \cos(2\, x) + \frac{1}{2}\, \_C1 \quad (3.17)$$

```
> plot([seq(subs(_C1=k,f),k=-2..2)],x=-Pi..3*Pi);
```




```
> # Maple hat unendlich viele Loesungen gefunden
> # dies sind aber noch nicht alle
> simplify(subs(x=n*Pi,f)) assuming n::integer;
```

$$0 \tag{3.18}$$

```
> simplify(subs(x=n*Pi,diff(f,x))) assuming n::integer;
```

$$0 \tag{3.19}$$

```
> # man kann den Stellen x=n*Pi differenzierbar von einem
  Loesungszweig auf den anderen wechseln
```